

UNIVERSITÉ LIBRE DE BRUXELLES

DÉPARTMENT D'INFORMATIQUE



INFO-F403 - INTRODUCTION TO LANGUAGE THEORY AND
COMPILING

Project Report – Part 2

Author:

Hakim BOULAHYA

Professor:

Gilles GEERAERTS

November 23, 2017

Contents

1 Grammar	2
1.1 Unproductive and unreachable symbols (a)	2
1.2 Priority and associativity of the operators (b)	2
1.2.1 Arithmetic expressions	2
1.2.2 Boolean expressions	3
1.3 Removing left recursion	3
1.4 Factorisation	4
1.5 Transformed Grammar	6
2 LL(1)	7
2.1 $\text{First}^1(\cdot)$ and $\text{Follow}^1(\cdot)$	7

1 Grammar

1.1 Unproductive and unreachable symbols (a)

In the given grammar, there is no unproductive and/or unreachable symbols.

1.2 Priority and associativity of the operators (b)

Note In this section, P&A refers to priority and associativity of the operators, AE to arithmetic expression and BE to boolean expression.

1.2.1 Arithmetic expressions

Since an arithmetic expression must always be process first before being compared to another one in a boolean expression, we will consider those two separately.

First let's consider the P&A of the arithmetic expressions. We have the following P&A:

-	right
*, /	left
+, -	left

And the following grammar:

```

<ExprArith>  → [VarName]
              → [Number]
              → (<ExprArith>)
              → - <ExprArith>
              → <ExprArith> <Op> <ExprArith>

<Op>         → +
              → -
              → *
              → /

```

As mention in the course page 111, an AE must be a *sum of products*, more specifically in our case a $\{sum, subtraction\}$ of $\{products, division\}$. We will use the same atom definition in the course, with **Number** as the constant rule and **VarName** as the id rule. The minus operator as a right associativity, meaning that it is always linked to the atom next to the operator, so we will set this operator directly as an atom rule.

Same thing goes for the parenthesis. The must be handled without considering the operators outside the parenthesis, so as an atom.

We have the following grammar results:

```

<ExprArith>  → <ExprArith> <SumSubOp> <ExprProd>
              → <ExprProd>
<ExprProd>   → <ExprProd> <ProdOp> <Atom>
              → <Atom>
<SumSubOp>   → +
              → -
<ProdOp>     → *
              → /
<Atom>       → [VarName]
              → [Number]
              → - <Atom>
              → (<ExprArith>)

```

1.2.2 Boolean expressions

For boolean expressions we have the following P&A:

not	right
>, <, >=, <=, =, <> /	left
and	left
or	left

And the following grammar:

<Cond>	→ <Cond> <BinOp> <Cond>
	→ not <SimpleCond>
	→ <SimpleCond>
<SimpleCond>	→ <ExprArith> <Comp> <ExprArith>
<BinOp>	→ and
	→ or
<Comp>	→ =
	→ >=
	→ >
	→ <=
	→ <
	→ <>

Following the same principle as for AE, we have here *disjunction of conjunctions of comparaisons*. By using the same mechanics as above, we have this grammar:

<Cond>	→ <Cond> or <ConjCond>
	→ <ConjCond>
<ConjCond>	→ <ConjCond> and <AtomCond>
	→ <AtomCond>
<AtomCond>	→ <SimpleCond>
	→ not <SimpleCond>
<SimpleCond>	→ <ExprArith> <Comp> <ExprArith>
<Comp>	→ =
	→ >=
	→ >
	→ <=
	→ <
	→ <>

1.3 Removing left recursion

For the new AE:

<ExprArith>	→ <ExprProd> <ExprArithPrime>
<ExprArithPrime>	→ <SumSubOp> <ExprProd> <ExprArithPrime>
	→ ε
<ExprProd>	→ <Atom> <ExprProdPrime>
<ExprProdPrime>	→ <ProdOp> <Atom> <ExprProdPrime>
	→ ε
<SumSubOp>	→ +
	→ -
<ProdOp>	→ *
	→ /
<Atom>	→ [VarName]
	→ [Number]
	→ - <Atom>
	→ (<ExprArith>)

For the new BE:

<Cond>	→ <ConjCond> <CondPrime>
<CondPrime>	→ or <ConjCond> <CondPrime>
	→ ε
<ConjCond>	→ <AtomCond> <ConjCondPrime>
<ConjCondPrime>	→ and <AtomCond> <ConjCondPrime>
	→ ε
<AtomCond>	→ <SimpleCond>
	→ not <SimpleCond>
<SimpleCond>	→ <ExprArith> <Comp> <ExprArith>
<Comp>	→ =
	→ >=
	→ >
	→ <=
	→ <
	→ <>

Those are the only rules where left-recursion appears, all the other rules are right-recursion or no recursion at all.

1.4 Factorisation

We can only factorize the following set of rules: < InstList >, < If >, < For >.

We have the following new rules:

<InstList>	→ <Instruction> <InstListSeq>
<InstListSeq>	→ ; <InstList>
	→ ε
<If>	→ if <Cond> then <Code> <IfSeq>
<IfSeq>	→ endif
	→ else <Code> endif
<For>	→ for [VarName] from <ExprArith> <ForOp>
	to <ExprArith> do <Code> done
<ForOp>	→ by <ExprArith>
	→ ε

1.5 Transformed Grammar

[1]	<Program>	→ begin <Code> end
[2]	<Code>	→ <InstList>
[3]		→ ϵ
[4]	<InstList>	→ <Instruction> <InstListSeq>
[5]	<InstListSeq>	→ ; <InstList>
[6]		→ ϵ
[7]	<Instruction>	→ <Assign>
[8]		→ <If>
[9]		→ <While>
[10]		→ <For>
[11]		→ <Print>
[12]		→ <Read>
[13]	<Assign>	→ [VarName] := <ExprArith>
[14]	<ExprArith>	→ <ExprProd> <ExprArithPrime>
[15]	<ExprArithPrime>	→ <SumSubOp> <ExprProd> <ExprArithPrime>
[16]		→ ϵ
[17]	<ExprProd>	→ <Atom> <ExprProdPrime>
[18]	<ExprProdPrime>	→ <ProdOp> <Atom> <ExprProdPrime>
[19]		→ ϵ
[20]	<SumSubOp>	→ +
[21]		→ -
[22]	<ProdOp>	→ *
[23]		→ /
[24]	<Atom>	→ [VarName]
[25]		→ [Number]
[26]		→ - <Atom>
[27]		→ (<ExprArith>)
[28]	<If>	→ if <Cond> then <Code> <IfSeq>
[29]	<IfSeq>	→ endif
[30]		→ else <Code> endif
[31]	<Cond>	→ <ConjCond> <CondPrime>
[32]	<CondPrime>	→ or <ConjCond> <CondPrime>
[33]		→ ϵ
[34]	<ConjCond>	→ <AtomCond> <ConjCondPrime>
[35]	<ConjCondPrime>	→ and <AtomCond> <ConjCondPrime>
[36]		→ ϵ
[37]	<AtomCond>	→ <SimpleCond>
[38]		→ not <SimpleCond>
[39]	<SimpleCond>	→ <ExprArith> <Comp> <ExprArith>
[40]	<Comp>	→ =
[41]		→ >=
[42]		→ >
[43]		→ <=
[44]		→ <
[45]		→ <>
[46]	<While>	→ while <Cond> do <Code> done
[47]	<For>	→ for [VarName] from <ExprArith> <ForOp> to <ExprArith> do <Code> done
[48]	<ForOp>	→ by <ExprArith>
[49]		→ ϵ
[50]	<Print>	→ print([VarName])
[51]	<Read>	→ read([VarName])

Figure 1: Transformed Grammar

2 LL(1)

2.1 $\text{First}^1(\cdot)$ and $\text{Follow}^1(\cdot)$

Variables	$\text{First}^1(\cdot)$	$\text{Follow}^1(\cdot)$
Program	begin	
Code	[VarName] if while for print read ϵ	end else endif done
InstList	[VarName] if while for print read	end else endif done
InstListSeq	; ϵ	end else endif done
Instruction	[VarName] if while for print read	; end else endif done
Assign	[VarName]	; end else endif done
ExprArith	[VarName] [Number] - (; end else endif done) = > = > < = < < > and or then do by to do
ExprArithPrime	+ - ϵ	$\text{Follow}^1(<\text{ExprArith}>)$
ExprProd	[VarName] [Number] - (+ - $\text{Follow}^1(<\text{ExprArith}>)$
ExprProdPrime	* / ϵ	+ - $\text{Follow}^1(<\text{ExprArith}>)$
SumSubOp	+ -	[VarName] [Number] - (
ProdOp	* /	[VarName] [Number] - (
If	if	; end else endif done
Atom	[VarName] [Number] - (* / + - $\text{Follow}^1(<\text{ExprArith}>)$
IfSeq	endif else	; end else endif done
Cond	[VarName] [Number] - (not	then do
CondPrime	or ϵ	then do
ConjCond	[VarName] [Number] - (not	or then do
ConjCondPrime	and ϵ	or then do
AtomCond	[VarName] [Number] - (not	and or then do
SimpleCond	[VarName] [Number] - (and or then do
Comp	= > = > < = < < >	[VarName] [Number] - (
While	while	; end else endif done
For	for	; end else endif done
ForOp	by ϵ	to
Print	print	; end else endif done
Read	read	; end else endif done